Math 173
Calculus I

Dr. Constant J. Goutziers
Department of Mathematics, Computer Science and Statistics
goutzicj@oneonta.edu

Lesson 2
Mathematical Models

Initializations

```
> restart;
  with(plots):
  with(stats):
```

2.1 Mathematical Models

Examples

Example 2.1.1

Example 4 on Page 30 of the Textbook
A ball is dropped from the upper observation deck of the CN tower, 450 m above the ground, and its height *h* above the ground is recorded at one-second intervals in the table below. Find a model to fit the data and use the model to predict the time at which the ball hits the ground.

Time	*Height*
0	450
1	445
2	431
3	408
4	375
5	332
6	279
7	216
8	143
9	61

Solution
First, we code the data and make a scatter plot.

```
> T:=[seq(k, k=0..9)];
  H:=[450,445,431,408,375,332,279,216,143,61];
```

$$T := [0, 1, 2, 3, 4, 5, 6, 7, 8, 9]$$

$$H := [450, 445, 431, 408, 375, 332, 279, 216, 143, 61] \quad \textbf{(2.1.1.1)}$$

The **zip** command allows us to create the points (t, h) by merging these two lists like a zipper.

```
> points:=zip((x,y)->[x,y], T, H);
```

$points := [[0, 450], [1, 445], [2, 431], [3, 408], [4, 375], [5, 332], [6, 279], [7, 216], [8, 143], [9, 61]]$ **(2.1.1.2)**

The **pointplot** routine can be used to visualize the data.

```
> pointplot(points);
```



Observe the data suggest that we use a quadratic model. The **fit** routine in the **stats** package is able to generate a quadratic leastsquares approximation to the data points.

```
> QLS:=evalf(fit[leastsquare[[t,h], h=a*t^2+b*t+c, {a,b,c}]
  ]([T, H]));
```

$$QLS := h = -4.901515152\, t^2 + 0.9621212121\, t + 449.3636364$$ **(2.1.1.3)**

The **display** command allows us to show the data points and the least squares approximation in one picture. To enhance the clarity of the picture we represent the data points by circles rather than points. Moreover, because the plot routine requires

$$-4.901515152\, t^2 + 0.9621212121\, t + 449.3636364$$

as an argument, we use the **rhs** (= right hand side) routine to extract this expression from the equation

$$h = -4.901515152\, t^2 + 0.9621212121\, t + 449.3636364$$

```
> p1:=pointplot(points, symbol=circle, symbolsize=15):
  p2:=plot(rhs(QLS), t=0..10):
  display([p1, p2], labels=[t,h]);
```



Observe that $h(t) = -4.901515152\, t^2 + 0.9621212121\, t + 449.3636364$ approximates the height of the ball at time *t*. The time at which the ball hits the ground is computed by setting $h(t)$ equal to zero.

```
> tv:=solve(rhs(QLS)=0, t);
```

$tv := -9.477247590, 9.673538162$ **(2.1.1.4)**

Of course the time of impact **toi** needs to be positive, so

```
> toi:=tv[2];
```

$toi := 9.673538162$ **(2.1.1.5)**